Санкт-Петербургский политехнический университет Петра Великого
Институт компьютерных наук и технологий
Высшая школа интеллектуальных систем и суперкомпьютерных технологий

Отчёт по лабораторной работе № 2

Дисциплина: Низкоуровневое программирование

Вариант 6.

Выполнил студент гр. 3530901/90002 _____ М.В. Дергачев
(подпись)

Принял преподаватель _____ Д.С. Степанов
(подпись)

“_” _____ 2021 г.

Санкт-Петербург
2021

Задача

1. Разработать программу для EDSAC, реализующую определенную вариантом задания функциональность, и предполагающую загрузчик Initial Orders 1. Массив (массивы) данных и другие параметры (преобразуемое число, длина массива, параметр статистики и пр.) располагаются в памяти по фиксированным адресам.

2. Выделить определенную вариантом задания функциональность в замкнутую (closed) подпрограмму, разработать вызывающую ее тестовую программу. Использовать возможности загрузчика Initial Orders 2. Адрес обрабатываемого массива данных и другие параметры передавать через ячейки памяти с фиксированными адресами.

Вариант задания

По варианту номер 5 необходимо реализовать нахождение медианы массива in-place.

Решение

Медиана – элемент массива, который находится ровно посередине после сортировки. Для выполнения поставленной задачи программа проходится по массиву, сравнивая каждое число с каждым, прибавляя к счетчику 1, если второе число больше текущего; вычитая 1, если число меньше текущего и не делая ничего, если они равны.

В конце каждого прохода проверяется значение счетчика, если он меньше 0 или больше 1 (количество чисел больше текущего меньше, чем количество чисел меньше текущего и наоборот, соответственно), то программа за место текущего числа берет следующее из массива и начинает проход заново, если значение счетчика равно 0 или 1 (0, когда в массиве нечетное количество элементов и 1, когда - четное), то медиана найдена и программа, записав ее в 0-ую ячейку, завершает работу.

Initial Orders 1

Листинг программы находится в приложении 1.



Рис. 1. Исходный массив {3; 0; 2; 1}



Рис. 2. Результат (1)



Рис. 3. Исходный массив {3; 0; 2; 1; 4; 4}



Рис. 4. Результат (2)

Initial Orders 2

С учётом возможностей загрузчика Initial Orders 2 была написана программа, листинг которой находится в приложении 1.

После была переписана программа из Initial Orders 1 с учётом директив Initial Orders 2.

Руководство программиста:

Разделы [address] [length] [array] отвечают за установку параметров для подпрограммы.

[sub] – подпрограмма нахождения медианы. Далее приведу пояснения по коду:

[0-1] Запись адреса возврата для выхода из подпрограммы.

[2-9] Запись параметров в выделенные для этого ячейки.

[10-13] Проверка «не пора ли закончить работу»

[14-15] Запись элемента массива в рабочую ячейку памяти

[16-19] Обновление счетчика, проверка «не прошли ли мы весь массив»

[20-33] Основной функционал: увеличение счетчика если M > N

[34-46] Проверяем, нашли ли мы медиану, готовимся к следующему шагу

[47-57] Обновляем адреса, готовимся к следующему шагу

[58-59] Выход из подпрограммы.

[60-63] Шаблоны команд подпрограммы.

[test prog] – тестовая программа, которая вызывает замкнутую подпрограмму.

Вывод

В ходе выполнения данной лабораторной работы был получен опыт программирования на EDSAC и работы с двумя его загрузчиками.

Приложение 1

Листинг initial orders 1. Файл прилагается к отчету.

```
      T 106 S                          [ начало...                                                    ]
      X 0 S                            [ для пошаговой отладки использовать Z 0 S                     ]
      T 0 S                            [ обнуляем аккум, загружаем в 0-ую ячейку ноль                 ]
      T 3 S                            [ загружаем в 3 ячейку ноль (самый важный счетчик)             ]
      A [len] 99 S                     [ Загружаем в аккум длину обрабатываемого массива              ] [i]
      T 1 S                            [ Записываем в ячейку 1                                        ]
      A [len] 99 S                     [ Загружаем в аккум длину обрабатываемого массива              ] [j]
      T 2 S                            [ Записываем в лчейку 2                                        ]
      A [addr] 100 S                   [ загружаем в аккум адрес 0-го элемента массива                ]
      L 0 L                            [ Сдвигаем аккум на 1 разряд влево                             ]
      A [<r1>] 51 S                    [ прибавляем код инструкции                                    ]
      T [<r1>] 51 S                    [ ставим на место инструкцию с новым адресом                   ]
[f:]  A [addr] 100 S             [43]  [ загружаем в аккум адрес 0-го элемента массива                ]
      L 0 L                            [ сдвигаем аккум на 1 разряд влево                             ]
      A [<r2>] 57 S                    [ прибавляем код инструкции                                    ]
      T [<r2>] 57 S                    [ ставим на место инструкцию с новым адресом                   ]
[loop:]A 1 S                     [47]  [ загружаем счетчик необработанных элементов массива (i)       ]
       S [one] 97 S                    [ вычитаем 1                                                   ]
       G [<exit>] 96 S                 [ если результат меньше 0, завершаем работу                    ]
       T 1 S                           [ ставим на место значение счетчика и обнуляем аккум (i)       ]
[r1:]  A 0 S                     [51]  [ загружаем в аккум значение из ячейки N                       ]
       T 0 S                           [ записываем в рабочую ячейку, обнуляем аккум                  ]
[loop2]     A 2 S [j]            [53]  [ загружаем в аккум значение счетчика j                        ]
            S [one] 97 S                [ вычитаем 1                                                   ]
            G 77 [<end loop2>] S        [ если прошли все элементы -> на выход из цикла                ]
            T 2 S                       [ ставим на место значение счетчика и обнуляем аккум (j)       ]
[r2:]       A 0 S                [57]  [ загружаем в аккум значение из ячейки M                       ] [array(j)]
            S 0 S                       [ вычитаем N-ый елемента                                       ] [array(j)-array(i)]
            G 67 [<f1>] S               [ прыжок в 67 ячейку, если N > M                               ]
            S 97 [one] S                [ вычитаем 1, чтобы проверить на равенство                     ]
            G 71 [<f2>] S               [ прыжок в 71 ячейку, если N == M                              ]
            T 5 S                       [ если N < M, очищаем аккум и идем дальше                      ]
            A 3 S                       [ \                                                            ]
            A 97 [one] S                [ |  самый важный счетчик++                                    ]
            T 3 S                       [ /                                                            ]
            E 71 [<f2>] S               [ пропускаем f1 и прыгаем в f2                                 ]
[f1:]       T 5 S                [67]  [ очищаем аккум                                                ]
            A 3 S                       [ \                                                            ]
            S 97 [one] S                [ |   самый важный счетчик--                                   ]
            T 3 S                       [ /                                                            ]
[f2:]       T 5 S                [71]  [ очищаем аккум                                                ]
            A 97 [one] S                [ загружаем в аккум 1                                          ]
            L 0 L                       [ сдвигаем на 1 разряд влево                                   ]
            A 57 [<r2>] S               [ прибавляем код инструкции, исполненной на предыдущем шаге    ]
            T 57 [<r2>] S               [ записываем сформированную инструкцию в память                ]
            E 53 [<loop2>] S            [ повторяем все операции; аккумулятор обнулен                  ]
[end loop2:]T 5 S                [77]  [ очищаем аккум                                                ]
      A 3 S                             [ загружаем значение счетчика из 3 ячейки                      ]
      G 84 [<f3>] S                     [если самый важный счетчик < 0, то -> (f3) (возврат в loop1)]
      S 97 [one] S                      [ \    иначе -> выходим (-> 96)                                ]
      G 96 [<exit>] S                   [ |                                                            ]
      S 97 [one] S                      [ |                                                            ]
      G 96 [<exit>] S                   [ /                                                            ]
[f3:] T 5 S                      [84]  [ очищаем аккум                                                ]
      A 97 [one] S                      [ загружаем в аккум 1                                          ]
      L 0 L                             [ сдвиг на 1 разряд влево                                      ]
      A 51 [<r1>] S                     [ прибавляем код инструкции, исполненной на предыдущем шаге    ]
      T 51 [<r1>] S                     [ записываем сформированную инструкцию в память                ]
      A 98  S                           [ загружаем в аккум команду с 0 адресом из строки 98           ]
      T 57 [<r2>] S                     [ ставим кк на место                                           ]
      T 3 S                             [ обнуляем самый важный счетчик                                ]
      A 99 [len] S                      [ загружаем длину массива в акум                               ]
      T 2 S                             [ записываем во 2 ячейку (loop2)                               ]
      E 43 [<f>] S                      [ возврат в цикл                                               ]
[end loop:] [exit:] O 0 S               [ конец...                                                     ]
Z 0 S                            [96]  [ останов                                                      ]
[one] P 0 L                      [97]  [ константа (1)                          ]
A 0 S                            [98]  [ команда для копирования                ]
[len]P 2 L                       [99]  [ длина массвиа (5)                      ]
[addr]P 50 L [ 101 ]            [100]  [ индекс первого элемента массива (101) ]
[array]P 1 L  [3]
[array]P 0 S  [0]
[array]P 1 S  [2]
[array]P 0 L  [1]
[array]P 2 S  [4]
```

Листинг initial orders 2. Файл прилагается к отчету.

```
T 56 K                          [ начало...                                                                  ]
GK                              [ "контрольная точка"                                                        ]
[sub]
[0:] A 3 F                      [ пролог: формируем код инструкции возврата в Acc                            ]
[1:] T 59 [<ret>] @             [ пролог: записываем инструкцию возврата                                     ]

[2:] A 6  [address] F           [ загружаем в аккум адрес 0-го элемента массива                              ]
[3:] A 62 [r1init:] @           [ прибавляем код инструкции с нулевым полем адреса                           ]
[4:] T 14 [<r1>] @              [ записываем сформированную инструкцию, обнуляем аккум                       ]

[5:] A 1 F                      [ загрузка в аккумулятор размера массива                                     ]
[6:] T 7 F                      [ запись сформированной инструкции, обнуление аккумулятора                   ]

   [f:]
[7:] A 6 [address] F            [ загружаем в аккум адреса 0-го элемента массива                             ]
[8:] A 63 [r2init:] @           [ прибавляем код инструкции с полем адреса 1                                 ]
[9:] T 20 [<r2>] @              [ записываем сформированную инструкцию, обнуляем аккум                       ]

 [loop:]                        [loop1-----------------------------                                          ]
[10:]  A 1 F                    [ загружаем счетчик необработанных элементов массива i                       ]
[11:]  S 61 [one] @             [ уменьшаем на 1                                                             ]
[12:]  G 58 [<exit>] @          [ если результат меньше 0, завершаем работу                                  ]
[13:]  T 1 F                    [ обновляем значение счетчика и обнуляем аккумулятор i                       ]
   [r1:]
[14:]  A 0 F                    [ загружаем в аккум значение из ячейки N                                     ]
[15:]  T 0 F                    [ записываем это значение в рабочую ячейку, обнуляем аккум                   ] [array(i)]

    [loop2]                     [loop2~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~]
[16:]    A 2 F                  [ загружаем в аккум значение счетчика j                                      ]
[17:]    S 61 [one] @
[18:]    G 40 [<end loop2>] @
[19:]    T 2 F                  [ ставим на место значение счётчика j                                        ]
   [r2:]
[20:]    A 0 F                  [ загружаем в аккумулятор значения из ячейки M                               ] [array(j)]
[21:]    S 0 F                  [ вычитанем N-ого елемент                                                    ] [array(j)-array(i)]

[22:]    G 30 [<f1>] @          [ прыжок если N > M                                                          ]
[23:]    S 61 [one] @
[24:]    G 34 [<f2>] @          [ прыжок если N == M                                                         ]

[25:]    T 5 F                  [ обнуляем аккум                                                             ]
[26:]    A 4 F                  [ \                                                                          ]
[27:]    A 61 [one] @           [ |    самый важный счетчик++                                                  ]
[28:]    T 4 F                  [ /                                                                          ]
[29:]    E 34 [<f2>] @          [ пропускаем f1 и прыгаме в f2                                               ]

   [f1:]
[30:]    T 5 F                  [ обнуляем аккум                                                             ]
[31:]    A 4 F                  [ \                                                                          ]
[32:]    S 61 [one] @           [ |    самый важный счетчик--                                                  ]
[33:]    T 4 F                  [ /                                                                          ]

   [f2:]
[34:]    T 5 F                  [ обнуляем аккум                                                             ]
[35:]    A 61 [one] @
[36:]    L 0 D                  [ сдвигаем на 1 разряд влево                                                 ]
[37:]    A 20 [<r2>] @          [ прибавляем код инструкции, исполненной на предыдущем шаге                  ]
[38:]    T 20 [<r2>] @          [ записываем сформированную инструкцию в память                              ]
[39:]    E 16 [<loop2>] @       [ повторяем все операции; аккумулятор обнулен                                ]
```

```
   [end loop2:]                 [~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~]
[40:] T 5 F                     [ обнуляем аккум                                                           ]
[41:] A 4 F                     [ загружаем значение счетчика из 4 ячейки                                  ]
[42:] G 47 [<f3>] @             [ если самый важный счетчик < 0, то -> (f3) (возврат в loop1)              ]
[43:] S 61 [one] @              [ \   иначе -> выходим (->58)                                              ]
[44:] G 58 [<exit>] @           [ |                                                                        ]
[45:] S 61 [one] @              [ |                                                                        ]
[46:] G 58 [<exit>] @           [ /                                                                        ]

   [f3:]
[47:] T 5 F
[48:] A 61 [one] @
[49:] L 0 D                     [ сдвигаем на 1 разряд влево                                               ]
[50:] A 14 [<r1>] @             [ прибавляем код инструкции, исполненной на предыдущем шаге                ]
[51:] T 14 [<r1>] @             [ записываем сформированную инструкцию в память                            ]

[52:] A 60 [index0] @           [ адрес нулевого элемента массива, запомненный при инициализации подпрограммы ]
[53:] T 6 F                     [ записываем адрес массива в ячейку 0, обнуляем аккумулятор                ]

[54:] T 4 F                     [обнуляем самый важный счетчик                                             ]
[55:] A 7 [length] F            [ размер массива, запомненный при инициализации подпрограммы               ]
[56:] T 2 F                     [ записываем в 2 ячеку (for loop2)                                         ]
[57:] E 7 [<f>] @
   [end loop:]                  [----------------------------                                              ]
   [exit:]
[58:] T 5 F                     [ обнуляем аккум                                                           ]
   [ret:]
[59:] E 0 F                     [ эпилог: инструкция возврата из подпрограммы                              ]
[60:] [index0] P 133 F
[61:] [one] P 0 D               [ константа 1                                                              ]
[62:]  [r1init:]  A 0 F         [ основа для формирования инструкции с меткой r1                           ]
[63:]  [r2init:]  A 0 F         [ основа для формирования инструкции с меткой r2                           ]



[test prog]
GK
[ 0:] X 0 F
[ 1:] A 11 [address] @          [ адрес массива                                                            ]
[ 2:] T 6 F                     [ запись адреса массива в ячейку 0, обнуление аккумулятора                 ]
[ 3:] T 4 F
[ 4:] A 12 [length] @           [ длина массива                                                            ]
[ 5:] T 1 F                     [ запись длины массива в ячейку 1, обнуление аккумулятора (i)              ]
[ 6:] A 12 [length] @           [ длина массива                                                            ]
[ 7:] T 2 F                     [ запись длины массива в ячейку 2, обнуление аккумулятора (j)              ]
[ 8:] A 8 @                     [ \ вызов                                                                  ]
[ 9:] G 56 [<sub>] F            [ / подпрограммы                                                           ]
[10:] Z 0 F                     [ останов                                                                  ]

[address:]
[11:]P 13 [<array>] @           [ адрес массива = <Начало программы>+12(см. ниже) ]

[length:]
[12:]P 2 F                      [ 15 ]

[13: array: ]

P 0 D  [1]
P 1 F  [2]
P 1 D  [3]
P 0 F  [0]

EZ PF  [ директива I02,переход к исполнению]
```